eman ta zabal zazu

Universidad del País Vasco
Euskal Herriko Unibertsitatea

Material de estudio

OCW 2018: *Utilizando Mathematica como apoyo al cálculo algebraico en los grados de Ingeniería*

Tema 4. Espacios vectoriales de dimensión finita

Equipo docente del curso
Martín Yagüe, Luis
Unzueta Inchaurbe, Aitziber
Arrospide Zabala, Eneko
García Ramiro, María Begoña
Soto Merino, Juan Carlos
Alonso González, Erik

Departamento de Matemática Aplicada
Escuela de Ingeniería de Bilbao, Edificio II-I

OpenCourseWare

TEMA 4. ESPACIOS VECTORIALES DE DIMENSIÓN FINITA

Notación

Se recuerda que, por definición, un vector es un elemento de un espacio vectorial.

Si $\mathbb{E}_n$ denota un espacio vectorial genérico de dimensión n todos sus vectores constan de n coordenadas (que son únicas respecto de la base de referencia considerada):

$$\forall \vec{v} \in \mathbb{E}_n : \vec{v} = (a_1, a_2, \ldots, a_n) \in \mathbb{E}_n$$

En este tema se trata la forma de realizar operaciones entre vectores en los siguientes espacios vectoriales:

- $\mathbb{R}^n$ de n-tuplas de números reales
- $\mathbb{P}_n(x)$ de los polinomios en la variable x de grado menor o igual que n
- $\mathbb{M}_{m\times n}(\mathbb{R})$ de las matrices reales de orden $m\times n$

Se concluye con el análisis de subespacios vectoriales.

Operaciones con vectores de $\mathbb{R}^n$

Nota

Se trabaja con elementos del espacio vectorial $\mathbb{R}^n$, es decir, n-tuplas de números reales que en *Mathematica* se representan mediante listas.

Debe tenerse en cuenta que si un vector de un espacio vectorial cualquiera de dimensión n se expresa con respecto a una base de dicho espacio se obtienen sus n coordenadas de forma que las operaciones explicadas en este apartado pueden servir para cualquier espacio vectorial siempre que se obtenga, previamente, una lista de las coordenadas de los vectores que se quieren operar.

Por este motivo, en el apartado se hace referencia a espacios vectoriales genéricos: $\mathbb{E}_n$.

Definición

Un vector se define escribiendo entre llaves sus elementos separados por comas.

Ejemplo. Asignación a variables de los vectores: $\vec{v}(1, 2, 3), \vec{w}(0, -2, 3) \in \mathbb{E}_3$

```
v = {1, 2, 3}
```

```
{1, 2, 3}
```

```
w = {0, -2, 3}
```

```
{0, -2, 3}
```

Ejemplo. Asignación a una variable del vector genérico: $\vec{u}(x, y, z) \in \mathbb{E}_3$

```
u = {x, y, z}
```

```
{x, y, z}
```

Suma

El operador utilizado es: **+**

Ejemplo. Sumas de los vectores definidos en los ejemplos anteriores.

```
u + v
```

```
{1 + x, 2 + y, 3 + z}
```

```
u + v // MatrixForm
```

$$\begin{pmatrix} 1 + x \\ 2 + y \\ 3 + z \end{pmatrix}$$

```
v + w
```

```
{1, 0, 6}
```

Producto por escalar

El operador utilizado es: *****

Ejemplo. Producto por escalar de los vectores definidos en los ejemplos anteriores.

```
{α * u, α * v} (* ∀α∈ℝ *)
```

```
{{x α, y α, z α}, {α, 2 α, 3 α}}
```

```
α * v /. α → 2 (* sustitución del parámetro por un valor *)
```

```
{2, 4, 6}
```

```
α = -3; α * v (* asignación de un valor al parámetro *)
```

```
{-3, -6, -9}
```

Combinación lineal

Se define como:

$$\sum_{i=1}^{n} \alpha_i \cdot \vec{v}_i = \alpha_1 \cdot \vec{v}_1 + \alpha_2 \cdot \vec{v}_2 + \ \ldots + \alpha_n \cdot \vec{v}_n \quad \text{donde} \quad \alpha_i \in \mathbb{R},\ \vec{v}_i \in \mathbb{E}_n \ \forall\, i \ = 1,\ 2,\ \ldots,\ n$$

Ejemplo. Combinaciones lineales de los vectores definidos en los ejemplos anteriores.

```
4 v - 6 w
```

```
{4, 20, -6}
```

```
Clear[α]
```

```
α * v + β * w
```

```
{α, 2 α - 2 β, 3 α + 3 β}
```

```
α * v + β * w /. {α → 2, β → -1}
```

```
{2, 6, 3}
```

```
α = -3; β = 2; α * v + β * w
```

```
{-3, -10, -3}
```

Operaciones con polinomios

Definición como funciones de una variable x

$\mathbb{P}_n(x)$ denota el espacio vectorial de los polinomios en la variable x de grado menor o igual que n con coeficientes reales.

Sus vectores (polinomios) pueden definirse como funciones de la variable x para operar con ellos.

Ejemplo. Definición de los polinomios: $p_1(x) = 2x^2 - 1,\ p_2(x) = -x^3 - 2x^2 - 3x + 1 \in \mathbb{P}_3(x)$

```
p1[x_] := 2 x^2 - 1; p2[x_] := -x^3 - 2 x^2 - 3 x + 1;
```

Suma algebraica

Operador : **+**

Ejemplo. Suma de los polinomios del ejemplo anterior.

```
q[x_] = p1[x] + p2[x]
```

$-3x - x^3$

Producto

Operador : *****

Ejemplo. Producto de los polinomios del ejemplo anterior.

```
r[x_] = p1[x] * p2[x]
```

$(-1 + 2x^2)(1 - 3x - 2x^2 - x^3)$

```
Expand[r[x]]
```

$-1 + 3x + 4x^2 - 5x^3 - 4x^4 - 2x^5$

Cociente

Operador : **/**

Ejemplo. Cociente de los polinomios $p_1(x)$ y $p_2(x)$.

```
s[x_] = p1[x] / p2[x]
```

$$\frac{-1 + 2x^2}{1 - 3x - 2x^2 - x^3}$$

```
ExpandAll[s[x]]
```

$$-\frac{1}{1 - 3x - 2x^2 - x^3} + \frac{2x^2}{1 - 3x - 2x^2 - x^3}$$

La siguientes funciones permiten obtener el cociente de dos polinomios y su resto:

- **PolynomialReduce**. Realiza el cociente de dos polinomios: *pol2*(x) entre *pol1*(x). La salida que produce es una lista **{{a},b}** donde **a** es el cociente y **b** el resto.
- **PolynomialQuotient**. Calcula el cociente de dos polinomios, *pol2*(x) entre *pol1*(x), eliminando el resto.
- **PolynomialRemainder**. Obtiene el resto del cociente de dos polinomios, *pol2*(x) entre *pol1*(x).

Ejemplo. Cálculo comparativo del cociente de dos polinomios con las funciones descritas.

```
PolynomialQuotient[p2[x], p1[x], x] (* cociente p2(x)/p1(x) *)
```

$-1 - \frac{x}{2}$

```
PolynomialRemainder[p2[x], p1[x], x]
```

$-\frac{7 x}{2}$

```
PolynomialReduce[p2[x], p1[x], x]
```

$\left\{\left\{-1 - \frac{x}{2}\right\}, -\frac{7 x}{2}\right\}$

```
(* Comprobación de la corrección del resultado obtenido en la salida anterior *)
```

$p2[x] == \text{Expand}[a * p1[x]] + b /. \left\{a -> -1 - \frac{x}{2}, b -> -\frac{7 x}{2}\right\} // \text{Simplify}$

```
True
```

Ejemplo. Cálculo comparativo del cociente de dos polinomios con las funciones descritas.

```
PolynomialQuotient[p1[x], p2[x], x] (* cociente p1(x)/p2(x) *)
```

```
0
```

```
PolynomialRemainder[p1[x], p2[x], x]
```

$-1 + 2 x^2$

```
PolynomialReduce[p1[x], p2[x], x]
```

$\{\{0\}, -1 + 2 x^2\}$

Escribir polinomios como listas

Puede obtenerse una lista formada por los elementos de un polinomio $\mathbb{P}_n(x)$ y operar con ella como si se tratase de un vector del espacio vectorial $\mathbb{R}^{n+1}$.

Se utiliza la función:

- **CoefficientList**. Devuelve una lista con los coeficientes de las potencias de la variable, x, del polinomio comenzando por la potencia cero (término independiente); permite, por tanto, obtener una representación vectorial (lista de coordenadas) de un polinomio con respecto a la base canónica del espacio.

Ejemplo. Obtener una lista con los coeficientes del polinomio $q_1(x) = -x^4 + 6x + 1 \in \mathbb{P}_4(x)$.

```
q1[x_] = 1 + 6 x - x^4
```

$1 + 6 x - x^4$

```
CoefficientList[q1[x], x]
```

```
{1, 6, 0, 0, -1}
```

Siempre debe tenerse en cuenta el espacio vectorial al que pertenece el polinomio considerado. Si un polinomio carece de términos de las potencias mayores el programa no devuelve un coeficiente nulo para cada una de ellas.
En ese caso, después de la variable, se añade a la función un argumento que indica el número de coordenadas del vector resultado.

Ejemplo. Sea el polinomio: $p_1(x) = 2\,x^2 - 1 \in \mathbb{P}_2\,(x) \subset \mathbb{P}_3(x) \subset \mathbb{P}_4\,(x)$

```
CoefficientList[p1[x], x] (* coeficientes de p1(x) como polinomio de ℙ2(x) *)
```

```
{-1, 0, 2}
```

```
CoefficientList[p1[x], x, 4] (* coeficientes de p1(x) como polinomio de ℙ3(x) *)
```

```
{-1, 0, 2, 0}
```

```
CoefficientList[p1[x], x, 5] (* coeficientes de p1(x) como polinomio de ℙ4(x) *)
```

```
{-1, 0, 2, 0, 0}
```

Operaciones con matrices

Recordatorio

Las operaciones con matrices se detallaron y fueron el objeto principal del *Tema 2*.

Escribir matrices como listas

Puede obtenerse una lista formada por los coeficientes de una matriz $\mathbb{M}_{m\times n}(\mathbb{R})$ y operar con ella como si se tratase de un vector del espacio vectorial $\mathbb{R}^{m\times n}$.

Se utiliza la función:

- **Flatten**. Convierte una matriz de orden $m\times n$ en un vector de $m\cdot n$ coordenadas. Permite, por tanto, obtener una representación vectorial (lista de coordenadas) de una matriz con respecto a la base canónica del espacio.

Ejemplo. Transformar una matriz 5×4 en un vector de $5\cdot 4 = 20$ coordenadas.

```
m1 = {{1, 0, 0, 0}, {-4, 0, 2, 3}, {3, 3, 3, 1}, {1, 0, 1, 0}, {-1, -3, 2, 1}};
```

```
Flatten[m1]
```

```
{1, 0, 0, 0, -4, 0, 2, 3, 3, 3, 3, 1, 1, 0, 1, 0, -1, -3, 2, 1}
```

Ejemplo. Obtener una lista con los coeficientes de los polinomios $q_1\,(x)$, $p_1\,(x)$, $p_2\,(x) \in \mathbb{P}_4\,(x)$ definidos en ejemplos anteriores.

```
Flatten[CoefficientList[{p1[x], p2[x], q[x]}, x, 5]]
```

```
{-1, 0, 2, 0, 0, 1, -3, -2, -1, 0, 0, -3, 0, -1, 0}
```

Subespacios vectoriales

Obtención de una base de un subespacio $U \subset \mathbb{E}_n$

Sea un subespacio U generado por un sistema S de vectores denotado de cualquiera de las dos formas siguientes:

$$U = \mathcal{L}\,(S) = \langle\, S \,\rangle$$

Los vectores se consideran como listas cuyos elementos son sus coordenadas referidas a una base del espacio vectorial considerado.

Procedimiento para obtener una base del subespacio U :

- los vectores del sistema S se disponen como filas de una matriz, M
- se obtiene la matriz escalonada equivalente a la matriz M (función **RowReduce**)
- las filas no nulas de la matriz escalonada devuelta constituyen una base de U
- el número de vectores de una base cualquiera es la dimensión de U : $dim(U)$

El rango de la matriz M indica el número de vectores libres del sistema generador S; por tanto, puede obtenerse una base diferente escogiendo un sistema libre a partir de S.

Ejemplo. Obtener una base de un subespacio $U \subseteq \mathbb{R}^4$ tal que:

$$U = \langle (1, 0, 0, 0), (-4, 0, 2, 3), (3, 3, 3, 1), (1, 0, 1, 0), (-1, -3, 2, 1) \rangle$$

- Introducción de la matriz 5×4

```
m1 = {{1, 0, 0, 0}, {-4, 0, 2, 3}, {3, 3, 3, 1}, {1, 0, 1, 0}, {-1, -3, 2, 1}};
```

- Obtención de la matriz escalonada equivalente

```
m = RowReduce[m1]
```

```
{{1, 0, 0, 0}, {0, 1, 0, 0}, {0, 0, 1, 0}, {0, 0, 0, 1}, {0, 0, 0, 0}}
```

```
MatrixForm[m]
```

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \\ 0 & 0 & 0 & 0 \end{pmatrix}$$

```
base1 = Table[m[[i]], {i, Length[m] - 1}]
```

```
{{1, 0, 0, 0}, {0, 1, 0, 0}, {0, 0, 1, 0}, {0, 0, 0, 1}}
```

- **Solución**. Base y dimensión del subespacio U:

$$\boxed{\begin{array}{c} dim(U) = 4 \Rightarrow U = \mathbb{R}^4 \\ B = \{ (1, 0, 0, 0), (0, 1, 0, 0), (0, 0, 1, 0), (0, 0, 0, 1) \} \end{array}}$$

Ejemplo. Obtener una base de un subespacio $U \subseteq \mathbb{R}^4$ tal que:

$$U = \langle (1, 0, 0, 0), (-4, 0, 2, 3), (3, 3, 3, 1), (0, 3, 5, 4) \rangle$$

- Introducción de la matriz 4×4:

```
m2 = {{1, 0, 0, 0}, {-4, 0, 2, 3}, {3, 3, 3, 1}, {0, 3, 5, 4}};
```

- Obtención de la matriz escalonada equivalente

```
m = RowReduce[m2]
```

$$\left\{\{1, 0, 0, 0\}, \left\{0, 1, 0, -\frac{7}{6}\right\}, \left\{0, 0, 1, \frac{3}{2}\right\}, \{0, 0, 0, 0\}\right\}$$

```
MatrixForm[m]
```

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & -\frac{7}{6} \\ 0 & 0 & 1 & \frac{3}{2} \\ 0 & 0 & 0 & 0 \end{pmatrix}$$

```
base1 = Table[m[[i]], {i, MatrixRank[m]}]
```

$\left\{\{1, 0, 0, 0\}, \left\{0, 1, 0, -\frac{7}{6}\right\}, \left\{0, 0, 1, \frac{3}{2}\right\}\right\}$

- **Solución**. Base y dimensión del subespacio U:

$$\boxed{\begin{array}{c} dim(U) = 3 \\ B_1 = \left\{(1, 0, 0, 0), \left(0, 1, 0, -\frac{7}{6}\right), \left(0, 0, 1, \frac{3}{2}\right)\right\} \end{array}}$$

- Obtención de una base a partir del sistema generador original

```
base2 = Table[m2[[i]], {i, MatrixRank[m2]}] (* se consideran los tres primeros vectores del sistema *)
```

{{1, 0, 0, 0}, {-4, 0, 2, 3}, {3, 3, 3, 1}}

```
MatrixRank[base2] (* son libres con lo que constituyen una base del subespacio *)
```

3

- **Solución**. Otra base del subespacio U:

$$\boxed{B_2 = \{(1, 0, 0, 0), (-4, 0, 2, 3), (3, 3, 3, 1)\}}$$

Cálculo de las ecuaciones implícitas de un subespacio $U \subset \mathbb{E}_n$

Se considera un subespacio $U \subset \mathbb{E}_n$; entonces $dim(U) = p < n = dim(\mathbb{E}_n)$.

Sea una base de U: $B_U = \{\vec{u}_1, \vec{u}_2, \ldots, \vec{u}_p\}$.

Procedimientos para obtener las ecuaciones implícitas del subespacio U :

- 1) cualquier vector $\vec{u} \in U$ puede expresarse como combinación lineal de los vectores de la base B_U; entonces, se resuelve el sistema: $\vec{u} = \{a_1 \cdot \vec{u}_1 + a_2 \cdot \vec{u}_2 + \ldots + a_p \cdot \vec{u}_p\}\ \forall\, \vec{u} \in U$
- 2) como $r(\vec{u}_1 \,|\, \vec{u}_2 \,|\, \ldots \,|\, \vec{u}_p \,|\, \vec{u}) = p\ \ \forall\, \vec{u} \in U$, se anulan los determinantes de las submatrices correspondientes

Ejemplo. Obtener la ecuación implícita del subespacio:

$$V = \langle (1, -1, 0), (-1, 0, 1), (-1, -2, 3) \rangle \subset \mathbb{R}^3$$

- Introducción de la familia generadora:

```
gv = {v1 = {1, -1, 0}, v2 = {-1, 0, 1}, v3 = {-1, -2, 3}}; MatrixRank[gv]
```

2

- Obtención de una base de V:

```
bv = {v1, v2}; MatrixRank[bv]
```

2

- Definición de un vector genérico $\vec{u} = (x, y, z) \in V$:

```
u = {x, y, z};
```

- Procedimiento 1:

```
Eliminate[a1 * v1 + a2 * v2 == u, {a1, a2}]
```

-y - z == x

- Procedimiento 2:

```
mat = Join[bv, {u}]; MatrixForm[mat]
```

$$\begin{pmatrix} 1 & -1 & 0 \\ -1 & 0 & 1 \\ x & y & z \end{pmatrix}$$

```
Solve[Det[mat] == 0, u]
```

Solve: Equations may not give solutions for all "solve" variables.

```
{{z → -x - y}}
```

- Solución: $\boxed{V = \{(x, y, z) \in \mathbb{R}^3 / x + y + z = 0\}}$

Ejemplo. Obtener una base del subespacio: $V = \{(x, y, z) \in \mathbb{R}^3 / x + y + z = 0\}$

- Dimensión de V: $dim\,(V) = dim\,(\mathbb{R}^3) - n^o\ ecuacione\ libres = 3 - 1 = 2$
- Definición de un vector genérico $\vec{u} = (x, y, z) \in V$:

```
u = {x, y, z};
```

- Relación entre coordenadas

```
s = Solve[x + y + z == 0, z]
```

```
{{z → -x - y}}
```

```
u2 = u /. s[[1]]
```

```
{x, y, -x - y}
```

- Obtención de la base

```
b = {u2 /. {x → 1, y → 0}, u2 /. {x → 0, y → 1}}
```

```
{{1, 0, -1}, {0, 1, -1}}
```

- Solución: $\boxed{B_V = \{(1, 0, -1), (0, 1, -1)\}}$

Coordenadas de un vector referidas a una base

Cualquier vector del espacio puede expresarse, por tanto, como combinación lineal de los vectores de una base cualquiera. Sus coordenadas son los coeficientes de dicha combinación lineal.

Ejemplo. Expresar las coordenadas de los vectores de la base B_1 del ejemplo anterior respecto de los vectores de la base B_2.

```
Solve[{1, 0, 0, 0} == a {1, 0, 0, 0} +b {-4, 0, 2, 3} +c {3, 3, 3, 1}, {a, b, c}]
```

```
{{a → 1, b → 0, c → 0}}
```

```
Solve[{0, 1, 0, -7/6} == a {1, 0, 0, 0} +b {-4, 0, 2, 3} +c {3, 3, 3, 1}, {a, b, c}]
```

```
{{a → -3, b → -1/2, c → 1/3}}
```

```
Solve[{0, 0, 1, 3/2} == a {1, 0, 0, 0} +b {-4, 0, 2, 3} +c {3, 3, 3, 1}, {a, b, c}]
```

```
{{a → 2, b → 1/2, c → 0}}
```

$$\boxed{\begin{array}{c}(1,\,0,\,0,\,0) = (\,1,\,0,\,0\,)_{B_2}\\ \left(0,\,1,\,0,\,-\frac{7}{6}\right) = \left(-3,\,-\frac{1}{2},\,\frac{1}{3}\right)_{B_2}\\ \left(0,\,0,\,1,\,\frac{3}{2}\right) = \left(2,\,\frac{1}{2},\,0\right)_{B_2}\end{array}}$$

Ejemplo. Expresar las coordenadas de los vectores de la base B_2 del ejemplo anterior respecto de los vectores de la base B_1.

```
Solve[{1, 0, 0, 0} == a {1, 0, 0, 0} + b {0, 1, 0, -7/6} + c {0, 0, 1, 3/2}, {a, b, c}]
{{a → 1, b → 0, c → 0}}

Solve[{-4, 0, 2, 3} == a {1, 0, 0, 0} + b {0, 1, 0, -7/6} + c {0, 0, 1, 3/2}, {a, b, c}]
{{a → -4, b → 0, c → 2}}

Solve[{3, 3, 3, 1} == a {1, 0, 0, 0} + b {0, 1, 0, -7/6} + c {0, 0, 1, 3/2}, {a, b, c}]
{{a → 3, b → 3, c → 3}}
```

$$\boxed{\begin{array}{c}(1,\,0,\,0,\,0) = (\,1,\,0,\,0\,)_{B_1}\\ (-4,\,0,\,2,\,3) = (-4,\,0,\,2)_{B_1}\\ (3,\,3,\,3,\,1) = (3,\,3,\,3)_{B_1}\end{array}}$$